

## SGPLOT Statements

### Proc statement

```
PROC SGPLOT <DATA= input-data-set>
  <CYCLEATTRS | NOCYCLEATTRS>
  <DESCRIPTION= "string">
  <NOAUTOLEGEND>
  <TMPLOUT= "filename">
  <UNIFORM= GROUP | SCALE | ALL>;
```

### Basic plots

```
BAND X=variable | Y=variable
  LOWER=number | numeric-variable
  UPPER=number | numeric-variable
  </ options >;
BUBBLE X=variable | Y=variable
  SIZE=numeric-variable </ options >;
HIGHLOW X=variable | Y=variable
  HIGH=numeric-variable
  LOW=numeric-variable </ options >;
NEEDLE X=variable Y=numeric-variable
  </ options >;
SCATTER X=variable Y=variable
  </ options >;
SERIES X=variable Y=variable
  </ options >;
STEP X=variable Y=numeric-variable
  </ options >;
VECTOR X=numeric-variable
  Y=numeric-variable </ options >;
```

### Distribution plots

```
DENSITY numeric-variable </ options >;
HBOX numeric-variable </ options >;
HISTOGRAM numeric-variable
  </ options >;
VBOX numeric-variable </ options >;
```

### Fit and confidence plots

```
LOESS X=numeric-variable
  Y=numeric-variable
  </ smoothing-options >
  < options >;
PBSPLINE X=numeric-variable
  Y=numeric-variable
```

```
</ smoothing-options >
  < options >;
REG X=numeric-variable
  Y=numeric-variable
  </ smoothing-options >
  < options >;
ELLIPSE X=numeric-variable
  Y=numeric-variable
  </ smoothing-options >
  < options >;
```

### Some common *smoothing-options*:

```
ALPHA= numeric-value
CLM <="text-string">
SMOOTH= numeric-value
WEIGHT= numeric-value
```

### Categorization plots

```
DOT category-variable </ options >;
HBAR category-variable </ options >;
HLINE category-variable </ options >;
VBAR category-variable </ options >;
VLINE category-variable </ options >;
```

### Common plot options

```
LEGENDLABEL="string"
NAME="string"
TRANSPARENCY=number
X2AXIS , Y2AXIS
```

### Axes and Reference lines

```
REFLINE value-list | variable
  </ options >;
XAXIS < options >;
X2AXIS < options >;
YAXIS < options >;
Y2AXIS < options >;
```

### Some common axis options

```
DISPLAY = ALL | NONE | (display-items)
  display-items = NOLABEL | NOLINE
  | NOTICKS | NOVALUES
GRID
LABEL = "string"
MAX = number , MIN = number
OFFSETMAX = number
OFFSETMIN = number
```

```
TICKVALUEFORMAT = DATA | sas-format
TYPE = LINEAR | LOG | TIME | DISCRETE
```

### Insets and Legends

```
INSET "string-1" ... "string-n"
  | ( "label-1" = "value-1"
  ... "label-n" = "value-n" )
  </ options >;
KEYLEGEND "plot-name-1" ... "plot-name-n"
  </ options >;
```

### Some **KEYLEGEND** options:

```
ACROSS = integer
BORDER | NOBORDER
DOWN = integer
LOCATION = OUTSIDE | INSIDE
POSITION = BOTTOM | TOP | RIGHT | LEFT
  | TOPRIGHT | TOPLEFT
  | BOTTOMRIGHT | BOTTOMLEFT
TITLE= "string"
```

### Also see SAS 9.3 doc on:

- **HBARPARM, VBARPARM**
- **LINEPARM**
- **WATERFALLPLOT (SGPLOT only)**
- **Discrete Attribute Maps**
- **Annotation**

### For more information, see:

#### Papers:

[http://support.sas.com/resources/papers/tnot/tnote\\_graph.html](http://support.sas.com/resources/papers/tnot/tnote_graph.html)

#### SAS<sup>®</sup> 9.3 documentation:

<http://support.sas.com/documentation/onlinedoc/graph/index.html>




## SGPLOT Procedure Tip Sheet

We've put all the information here to get you started with the SGPLOT procedure. The examples on the reverse side can be typed into the program editor and run.

This procedure lets you quickly create single-celled graphs with scatter plots, series plots, vector plots, confidence bands, prediction or confidence ellipses, fit lines, histograms, density plots, dot plots, bar charts, box plots and many more.

The SG procedure family emphasizes good default behavior that lends itself well to effective graphics. These procedures are based on the Graph Template Language (GTL) and fit into the ODS Graphics.



```
proc sgplot data=sashelp.heart;
  title "Cholesterol Distribution";
  histogram cholesterol;
  density cholesterol;
  density cholesterol / type=kernel;
run;
```




SAS Institute Inc. World Headquarters  
+1 919 677 8000 [www.sas.com/offices](http://www.sas.com/offices)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2012, SAS Institute Inc. All rights reserved. 543371\_0612


# SAS<sup>®</sup> 9 SGPlot Procedures Tip Sheet

## SGPlot: Basic Series with Band




```
proc sgplot data=sashelp.stocks
  (where=(stock='IBM'));
  band x=date upper=high lower=low /
  legendLabel="High-Low";
  series x=date y=close /
  lineattrs=(pattern=dot);
run;
```

## SGPlot: Vertical Box with footnote




```
proc sgplot data=sashelp.cars
  (where=(origin='USA'));
  vbox mpg_city / category=type;
  footnote height=1 justify=right
  "Created at: &system";
run;
```

## SGPlot: Dot plot




```
proc sgplot data=sashelp.cars
  (where=(origin='Europe'));
  dot make / response=mpg_city
  stat=mean limitstat=clm;
run;
```

## SGPlot: Modify axis




```
proc sgplot data=sashelp.class;
  vector x=weight y=height / datalabel
  xorigin=100 yorigin=62.3 group=sex;
  yaxis grid;
  xaxis label="Student Weight" grid;
run;
```

## SGPlot: Horizontal Box with title




```
proc sgplot data=sashelp.class;
  title "Student Weight distribution";
  hbox weight / category=sex ;
run;
```

## SGPlot: Loess fit




```
proc sgplot data=sashelp.class;
  loess x=age y=height / group=sex clm
  clmtransparency=0.6;
run;
```

## SGPlot: BY groups



```
proc sgplot data=sashelp.cars(where=
  (make in ('Acura','Volvo')));
  by make;
  vbar type / response=mpg_city
  group=drivetrain stat=mean;
run;
```

## SGPlot: Modify legend



```
proc sgplot data=sashelp.class;
  ellipse x=weight y=height / fill
  alpha=0.2;
  scatter x=weight y=height /
  group=sex name="sp1";
  keyLegend "sp1"/ title="Gender:"
  location=inside;
run;
```